

Science and Technology Facilities Council

GALAHAD
SCU

## 1 SUMMARY

GALAHAD_SCU is a suite of Fortran procedures for computing the the solution to an extended system of $n+m$ sparse real linear equations in $n+m$ unknowns,

$$
\left(\begin{array}{ll}
\mathbf{A} & \mathbf{B}  \tag{1.1}\\
\mathbf{C} & \mathbf{D}
\end{array}\right)\binom{\mathbf{x}_{1}}{\mathbf{x}_{2}}=\binom{\mathbf{b}_{1}}{\mathbf{b}_{2}} .
$$

in the case where the $n$ by $n$ matrix $\mathbf{A}$ is nonsingular and solutions to the systems

$$
\mathbf{A x}=\mathbf{b} \text { and } \mathbf{A}^{T} \mathbf{y}=\mathbf{c}
$$

may be obtained from an external source, such as an existing factorization. The subroutine uses reverse communication to obtain the solution to such smaller systems. The method makes use of the Schur complement matrix

$$
\mathbf{S}=\mathbf{D}-\mathbf{C A}^{-1} \mathbf{B} .
$$

The Schur complement is stored and factorized as a dense matrix and the subroutine is thus appropriate only if there is sufficient storage for this matrix. Special advantage is taken of symmetry and definiteness in the coefficient matrices. Provision is made for introducing additional rows and columns to, and removing existing rows and columns from, the extended matrix.

ATTRIBUTES - Versions: GALAHAD_SCU_single, GALAHAD_SCU_double, Uses: _ROT, _ROTG. Date: October 2001. Origin: N. I. M. Gould, Rutherford Appleton Laboratory, and Ph. L. Toint, University of Namur, Belgium. Language: Fortran 95 + TR 15581 or Fortran 2003

## 2 HOW TO USE THE PACKAGE

The package is available using both single and double precision reals, and either 32-bit or 64-bit integers. Access to the 32-bit integer, single precision version requires the USE statement

```
USE GALAHAD_SCU_single
```

with the obvious substitution GALAHAD_SCU_double, GALAHAD_SCU_single_64 and GALAHAD_SCU_double_64 for the other variants.
If it is required to use more than one of the modules at the same time, the derived types SCU_matrix_type, SCU_inform_type, and SCU_data_type (Section 2.2), and the subroutines SCU_factorize, SCU_solve, SCU_append, SCU_delete, and SCU_terminate (Section 2.3) must be renamed on one of the USE statements.

### 2.1 Real and integer kinds

We use the terms integer and real to refer to the fortran keywords REAL (rp_) and INTEGER (ip_), where rp_ and $i p_{\text {_ }}$ are the relevant kind values for the real and integer types employed by the particular module in use. The former are equivalent to default REAL for the single precision versions and DOUBLE PRECISION for the double precision cases, and correspond to rp_ = real32 and rp_ = real64, respectively, as supplied by the fortran iso_fortran_env module. The latter are default (32-bit) and long (64-bit) integers, and correspond to ip_ = int 32 and ip_ = int 64, respectively, again from the iso_fortran_env module.

[^0]
### 2.2 The derived data types

Three derived data types are accessible from the package.

### 2.2.1 The derived data type for holding matrix information

The derived data type SCU_matrix_type is used to hold data about the extended matrix and its factors. The components of SCU_matrix_type are:
$n \quad$ is a scalar of type INTEGER (ip_), that holds the value of $n$, the dimension of the matrix $\mathbf{A}$.
$m \quad$ is a scalar of type INTEGER (ip_), that holds the value of $m$, dimension of the matrix $\mathbf{D}$.
m_max is an scalar of type INTEGER (ip_), that holds the dimension of the largest possible matrix $\mathbf{D}$ to be permitted by the package.
class is a scalar of type INTEGER (ip_), that indicates the type of matrix to be processed. Permitted values are:
1 the extended matrix is unsymmetric,
2 the extended matrix is symmetric,
3 the extended matrix is symmetric and the Schur complement matrix $\mathbf{S}$ is known to be positive definite.
4 the extended matrix is symmetric and the Schur complement matrix $\mathbf{S}$ is known to be negative definite.
BD_val is a rank-one allocatable of type REAL (rp_), that holds the values of the entries in the partitioned matrix

$$
\begin{equation*}
\binom{\mathbf{B}}{\mathbf{D}_{U}}, \tag{2.1}
\end{equation*}
$$

where $\mathbf{D}_{U}$ is the upper triangular part of $\mathbf{D}$, ie, $\left(\mathbf{D}_{U}\right)_{i j}=(\mathbf{D})_{i j}$ if $i \leq j$ and zero otherwise. The entries must be ordered by columns, with the entries in each column contiguous and those of column $j$ preceding those of column $j+1(j=1, \ldots, m)$. The ordering within each column is unimportant.

BD_row is a rank-one allocatable of type $\operatorname{INTEGER}\left(i p_{-}\right)$, that holds the row indices of the corresponding entries in BD_val.

BD_col_start is a rank-one allocatable of type INTEGER (ip_), that must be set so that BD_col_start (j) holds the positions in the arrays BD_val and BD_row of the first entry in column $j(j=1, \ldots, m)$. BD_col_start ( $\mathrm{m}+1$ ) must be set to the number of entries in the matrix (2.1) plus one.

CD_val is a rank-one allocatable of type default REAL (rp_), that need not be set in the symmetric case. In the unsymmetric case, it holds the values of the entries in the partitioned matrix

$$
\begin{equation*}
\left(\mathbf{C} \mathbf{D}_{L}\right), \tag{2.2}
\end{equation*}
$$

where $\mathbf{D}_{L}$ is the strict lower triangular part of $\mathbf{D}$, ie, $\left(\mathbf{D}_{L}\right)_{i j}=(\mathbf{D})_{i j}$ if $i>j$ and zero otherwise. The entries must be ordered by rows, with the entries in each row contiguous and those of row $i$ preceding those of row $i+1$ $(i=1, \ldots, m)$. The ordering within each row is unimportant.

CD_col is a rank-one allocatable of type INTEGER (ip_), that need not be set in the symmetric case. In the unsymmetric case, it holds the column indices of the corresponding entries in CD_val.

CD_row_start is a rank-one allocatable of type $\operatorname{INTEGER(ip_{-})\text {,thatneednotbesetinthesymmetriccase.Inthe}}$ unsymmetric case, it must be set so that CD_row_start (i) points to the positions in the arrays CD_val and CD_col of the first entry in row $i(i=1, \ldots, m)$. CD_row_start ( $\mathrm{m}+1$ ) must be set to the number of entries in the matrix (2.2) plus one.

## All use is subject to the conditions of a BSD-3-Clause License. <br> See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

### 2.2.2 The derived data type for holding problem data

The derived data type SCU_data_type is used to hold all the data for a particular problem between calls of SCU procedures. This data should be preserved, untouched, from the initial call to SCU_factorize to the final call to SCU_terminate. All components of SCU_data_type are private.

### 2.2.3 The derived data type for holding informational parameters

The derived data type SCU_inform_t ype is used to hold parameters that give information about the factorization. The components of SCU_inform_type are:
alloc_status is a scalar of type $\operatorname{INTEGER}\left(\mathrm{ip}_{-}\right)$, that contains the return status from the last attempted internal workspace array allocation or deallocation. A non-zero value indicates that the allocation or deallocation was unsuccessful, and corresponds to the STAT = value on the user's system. Consult local compiler documentation for further details.
inertia is a rank-one array of length 3 and type $\operatorname{INTEGER}$ (ip_), that holds the inertia of $\mathbf{S}$ when the extended matrix is symmetric. Specifically, inertia(i), $i=1,2,3$, give the number of positive, negative and zero eigenvalues of $\mathbf{S}$ respectively.

For backward compatibility with an earlier version of the package, there is an equivalent type SCU_info_type, and the two names may be used interchangeably.

### 2.3 Argument lists and calling sequences

To solve the extended system, the user must first call SCU_factorize to form and factorize the Schur complement, and then call SCU_solve to compute the solution to the extended system. The solution of additional extended systems, with the same coefficient matrix but different right-hand sides, may be found by further calls to SCU_solve.

The solution of further-extended systems of the form

$$
\left(\begin{array}{ccc}
\mathbf{A} & \mathbf{B} & \mathbf{c}_{1} \\
\mathbf{C} & \mathbf{D} & \mathbf{c}_{2} \\
\mathbf{r}_{1}^{T} & \mathbf{r}_{2}^{T} & d
\end{array}\right)\left(\begin{array}{c}
\mathbf{x}_{1} \\
\mathbf{x}_{2} \\
x_{n+m+1}
\end{array}\right)=\left(\begin{array}{c}
\mathbf{b}_{1} \\
\mathbf{b}_{2} \\
b_{n+m+1}
\end{array}\right) .
$$

may be found by firstly calling SCU_append to update the existing factorization of $\mathbf{S}$ (obtained from SCU_factorize) to give that of the Schur complement of $\mathbf{A}$ in the further-extended coefficient matrix and then by calling SCU_solve. Likewise, the solution of extended systems of the form

$$
\left(\begin{array}{cc}
\mathbf{A} & \overline{\mathbf{B}} \\
\overline{\mathbf{C}} & \overline{\mathbf{D}}
\end{array}\right)\binom{\mathbf{x}_{1}}{\mathbf{x}_{2}}=\binom{\mathbf{b}_{1}}{\mathbf{b}_{2}} .
$$

in which a row and column are removed from the coefficient matrix of (1.1), may be found by firstly calling SCU_delete to update the existing factorization of $\mathbf{S}$ (obtained from SCU_factorize) to give that of the Schur complement of $\mathbf{A}$ in the new extended coefficient matrix and then once again by calling SCU_solve.

Finally, the user may call SCU_terminate to deallocate any workspace used to hold the factors of $\mathbf{S}$.
We use square brackets [ ] to indicate OPTIONALarguments.

### 2.3.1 The factorization stage

The Schur complement matrix may be factorized as follows:

```
CALL SCU_factorize( matrix, data, VECTOR, status, info )
```

[^1]matrix is a scalar INTENT (INOUT) argument of type SCU_matrix_type (see Section 2.2.1). The following components are used by SCU_factorize:
n must be set to $n$, the dimension of the matrix $\mathbf{A}$. It is unchanged by the subroutine. Restriction: $\mathrm{n} \geq 0$.
m_max must be set to the dimension of the largest matrix $\mathbf{D}$ that will be allowed by the package. Internal workspace of dimension $m \_\max *\left(m \_\max +3\right) / 2\left(\right.$ or $3 m \_\max *\left(m \_\max +1\right) / 2$ if class $\left.=1\right)$ will be used by the package. Thus care should be taken not to set m_max larger than is absolutely necessary to account for the sequence of extended matrices likely to be encountered. Restriction: m_max $\geq 0$.
$m$ must be set to $m$, dimension of the matrix $\mathbf{D}$. It is unchanged by the subroutine. Restriction: $0 \leq \mathrm{m} \leq \mathrm{m}$ _max.
class must be set to indicate the type of matrix that will be processed (see Section 2.2.1). If $\mathbf{S}$ is known to be (positive or negative) definite throughout the sequence of extended matrices to be considered, it is more efficient to set class $=3$ or 4 . Likewise, if the matrix is known to be symmetric throughout the sequence of extended matrices to be considered, it is more efficient to set class $>1$. Restriction: $1 \leq$ class $\leq 4$.
BD_val and BD_row must be set as described in Section 2.2.1 to hold the values and row indices, respectively, of the matrix (2.1). Before use, the arrays must be ALLOCATEd to be of sufficient length to hold any matrix of the form (2.1) to be encountered in the sequence of extended matrices to be considered. Any elements in BD_val and BD_row that lie below the diagonal of $\mathbf{D}$ will be removed by SCU_factorize.

BD_col_start must be set as described in Section 2.2.1 to hold the positions of the start of the columns of (2.1), as well as to the first position past the end of the last column. Before use, this array must be ALLOCATEd to be of length at least m_max +1 . The values of this array may be altered if entries that lie below the diagonal of $\mathbf{D}$ are removed by SCU_factorize.

CD_val and CD_col. When class=1, these must be set as described in Section 2.2.1 to hold the values and column indices, respectively, of the matrix (2.2). Before use, the arrays must be ALLOCATEd to be of sufficient length to hold any matrix of the form (2.2) to be encountered in the sequence of extended matrices to be considered. Any elements in CD_val and CD_col that lie on or to the right of the diagonal of $\mathbf{D}$ will be removed by SCU_factorize. This component need not be ALLOCATEd or set if class $>1$.
CD_row_start . When class=1, this must be set as described in Section 2.2.1 to hold the positions of the start of the rows in (2.2), as well as to the first position past the end of the last row. Before use, this array must be ALLOCATEd to be of length at least m_max +1 . The values of this array may be altered if entries that lie on or to the right of the diagonal of $\mathbf{D}$ are removed by SCU_factorize. This component need not be ALLOCATEd or set if class $>1$.
data is a scalar INTENT (INOUT) argument of type SCU_data_type. It must not have been altered since the last call to SCU_factorize, SCU_append or SCU_delete.

VECTOR is an INTENT (INOUT) rank-one array argument of length matrix $\circ \mathrm{n}$ and type REAL (rp_), that needs not be set by the user on initial (status=1) entry. If status is greater than 1 on exit, a re-entry must be made with VECTOR set appropriately (see Section 2.4).
status is an scalar INTENT (INOUT) argument of type default INTEGER, that must be set by the user on initial input to 1 . On output, the value of status is used to request additional information, to signal an error in the input data or to indicate a successful call to the subroutine. A successful call is indicated by the exit value status=0. For other values, see Sections 2.4 and 2.5.
info is a scalar INTENT (INOUT) argument of type SCU_inform_type (see Section 2.2.3).

### 2.3.2 The solution stage

Solve the extended system of equations using the factorization produced by a previous call to SCU_factorize, SCU_append or SCU_delete, as follows:

```
CALL SCU_solve( matrix, data, RHS, X, VECTOR, status )
```


## All use is subject to the conditions of a BSD-3-Clause License. <br> See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

matrix is a scalar INTENT (IN) argument of type SCU_matrix_type (see Section 2.2.1), that must not have been changed since the last call to SCU_factorize, SCU_append or SCU_delete.
data is a scalar INTENT (INOUT) argument of type SCU_data_type. It must not have been altered since the last call to SCU_factorize, SCU_append or SCU_delete.

RHS is an INTENT (IN) rank-one array argument of length matrix $\%$ n+matrix $\%$ and type REAL (rp_), that must be set on entry to the values of the right-hand-side vector

$$
\binom{\mathbf{b}_{1}}{\mathbf{b}_{2}}
$$

of the extended system of equations. RHS is not altered by the subroutine.
$X$ is an INTENT (OUT) rank-one array argument of length matrix $\circ n+$ matrix $\circ m$ and type REAL (rp_). On final (status=0) exit, $X$ contains the values of the solution

$$
\binom{\mathbf{x}_{1}}{\mathbf{x}_{2}}
$$

to the extended system of equations.
VECTOR is an INTENT (INOUT) rank-one array argument of length matrix\%n and type REAL (rp_), that needs not be set by the user on initial (status=1) entry. If status is greater than 1 on exit, a re-entry must be made with VECTOR set appropriately (see Section 2.4).
status is an scalar INTENT (INOUT) argument of type default INTEGER, that must be set by the user on initial input to 1 . On output, the value of status is used to request additional information, to signal an error in the input data or to indicate a successful call to the subroutine. A successful call is indicated by the exit value status $=0$. For other values, see Section 2.4.

### 2.3.3 The updating stage

Call SCU_append to extend the factorization of the Schur complement when a new row and column are appended to the extended matrix. Subsequent systems of equations with the larger coefficient matrix may then be solved by calls to SCU_solve. Note in particular that the arrays matrix_CD_val, matrix_CD_col, matrix_CD_row_start, matrix_BD_val, matrix_BD_row and matrix_BD_col_start must be sufficiently large to allow for the incoming row and column. The factorization may be extended as follows:

```
CALL SCU_append( matrix, data, VECTOR, status, info )
```

matrix is a scalar INTENT (INOUT) argument of type SCU_matrix_type (see Section 2.2.1). The following components are used by SCU_append:
n , m_max, $m$ and class. These should not have been changed since the last call to SCU_factorize, SCU_append or SCU_delete. On a successful exit (status $=0$ ) from SCU_append, $n$ will be unaltered and $m$ will have been increased by 1 to account for the appended row and column. Restriction: $0 \leq m \leq m \_m a x-1$.
BD_val and BD_row. These must be set as described in Section 2.2.1 to hold the values and row indices, respectively, of the further extended matrix

$$
\left(\begin{array}{cc}
\mathbf{B} & \mathbf{c}_{1}  \tag{2.3}\\
\mathbf{D}_{U} & \mathbf{c}_{2} \\
& d
\end{array}\right)
$$

Specifically, the first col_start ( $m+1$ ) -1 elements of BD_val and BD_row should not have been changed since the last call to SCU_factorize, SCU_append or SCU_delete, while elements col_Start (m+1), ...

[^2]should contain the values and row indices of the appended column
\[

\left($$
\begin{array}{c}
\mathbf{c}_{1} \\
\mathbf{c}_{2} \\
d
\end{array}
$$\right)
\]

the ordering within this column being unimportant.
BD_col_start must be set as described in Section 2.2.1 to hold the positions of the start of the columns of (2.3), as well as to the first position past the end of the last column. Specifically, the first $m+1$ elements of BD_col_start should not have been changed since the last call to SCU_factorize, SCU_append or SCU_delete, while BD_col_Start ( $\mathrm{m}+2$ ) should be set to the number of entries in (2.3) plus one.
CD_val and CD_col. When SCU_factorize was last called with matrix\%class=1, these must be set as described in Section 2.2.1 to hold the values and column indices respectively of the further extended matrix

$$
\left(\begin{array}{cc}
\mathbf{C} & \mathbf{D}_{L}  \tag{2.4}\\
\mathbf{r}_{1}^{T} & \mathbf{r}_{2}^{T}
\end{array}\right) .
$$

Specifically, the first row_start ( $\mathrm{m}+1$ )-1 elements of CD_val and CD_col should not have been changed since the last call to SCU_factorize, SCU_append or SCU_delete, while elements row_start ( $\mathrm{m}+1$ ) , ... should contain the values and column indices of the appended row

$$
\left(\begin{array}{ll}
\mathbf{r}_{1} & \mathbf{r}_{2}^{T}
\end{array}\right)
$$

the ordering within this row being unimportant. This component need not be ALLOCATEd or set if SCU_factorize was last called with matrix\%class $>1$.
CD_row_start. When SCU_factorize was last called with matrix\%class=1, this must be set as described in Section 2.2.1 to hold the positions of the start of the rows of (6), as well as to the first position past the end of the last row. Specifically, the first $m+1$ elements of CD_row_start should not have been changed since the last call to SCU_factorize, SCU_append or SCU_delete, while CD_row_start ( $\mathrm{m}+2$ ) should be set to the number of entries in (2.4) plus one. This component need not be ALLOCATEd or set if SCU_factorize was last called with matrix\%class $>1$.
data is a scalar INTENT (INOUT) argument of type SCU_data_type. It must not have been altered since the last call to SCU_factorize, SCU_append or SCU_delete.

VECTOR is an INTENT (INOUT) rank-one array argument of length matrix\%n and type REAL (rp_), that needs not be set by the user on initial (status=1) entry. If status is greater than 1 on exit, a re-entry must be made with VECTOR set appropriately (see Section 2.4).
status is an scalar INTENT (INOUT) argument of type default INTEGER, that must be set by the user on initial input to 1 . On output, the value of status is used to request additional information, to signal an error in the input data or to indicate a successful call to the subroutine. A successful call is indicated by the exit value status=0. For other values, see Section 2.4 and 2.5.
info is a scalar INTENT (INOUT) argument of type SCU_inform_type (see Section 2.2.3).

### 2.3.4 The deletion stage

Call SCU_delete to extend the factorization of the Schur complement when a row and column of the existing extended matrix are to be removed. Subsequent systems of equations with the smaller coefficient matrix may then be solved by calls to SCU_solve. The data structures for holding B, C and $\mathbf{D}$ will be automatically updated to account for the row and column deletions. Compute the factorization of the extended matrix following a row and column removal as follows:

> CALL SCU_delete( matrix, data, VECTOR, status, info, col_del [, row_del ] )

## All use is subject to the conditions of a BSD-3-Clause License. <br> See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

matrix is a scalar INTENT (INOUT) argument of type SCU_matrix_type (see Section 2.2.1). The following components are used by SCU_delete :
n , m_max, m, class, BD_val, BD_row, BD_col_start, CD_val, CD_col and CD_row_start. These should not have been changed since the last call to SCU_factorize, SCU_append or SCU_delete. On a successful exit (status $=0$ ) from SCU_delete, $n$ will be unchanged, $m$ will have been decreased by 1 to account for the deleted row and column, and the contents of the arrays may have been changed.
data is a scalar INTENT (INOUT) argument of type SCU_data_type. It must not have been altered since the last call to SCU_factorize, SCU_append or SCU_delete.

VECTOR is an INTENT (OUT) rank-one array argument of length matrix\%n and type REAL (rp_), that is used as workspace.
status is an scalar INTENT (OUT) argument of type default INTEGER (ip_). On output, the value of status is used to signal an error in the input data or to indicate a successful call to the subroutine. A successful call is indicated by the exit value status=0. For other values, see Section 2.5.
info is a scalar INTENT (INOUT) argument of type SCU_inform_type (see Section 2.2.3).
col_del is an scalar INTENT (IN) argument of type INTEGER (ip_). On input, this must be set by the user to the index of the column of (2.1) that is to be removed. Restriction: $1 \leq$ col_del $\leq \mathrm{m}$.
row_del is an optional scalar INTENT (IN) argument of type INTEGER(ip_). If SCU_factorize was last called with matrix\%class=1, this can be set by the user to the index of the row of (2.2) that is to be removed. If the argument is absent, or if matrix\%class $>1$, row col_del will be removed. Restriction: $1 \leq$ row_del $\leq m$.

### 2.3.5 The final stage

Deallocate the space required to hold the factors of the (sequence of) Schur complement(s) as follows:

```
CALL SCU_terminate( data, status, info )
```

data is a scalar INTENT (INOUT) argument of type SCU_data_type. It must not have been altered since the last call to SCU_factorize, SCU_append or SCU_delete.
status is an scalar INTENT (OUT) argument of type default INTEGER, that contains the exit status following a call to SCU_terminate. A successful call is indicated by the exit value status=0. For other values, see Section 2.5 .
info is a scalar INTENT (OUT) argument of type SCU_inform_type (see Section 2.2.3).

### 2.4 Reverse communication

If one of the subroutines SCU_factorize, SCU_solve or SCU_append returns with a strictly possible value of status, the subroutine in question requires the user to manipluate the vector VECTOR, and to re-enter with status and all other arguments unchanged. Possible values of status and their consequences are as follows:

2 The user must obtain the solution to the system of equations $\mathbf{A x}=\mathbf{b}$. The particular vector $\mathbf{b}$ is returned in the array VECTOR; the user must calculate $\mathbf{x}$ and pass this vector back in VECTOR.

3 The user must obtain the solution to the system of equations $\mathbf{A}^{T} \mathbf{y}=\mathbf{c}$. The particular vector $\mathbf{c}$ is returned in the array VECTOR; the user must calculate $\mathbf{y}$ and pass this vector back in VECTOR.

[^3]
### 2.5 Warning and error messages

A negative value of status on exit from any of the preceding subroutines indicates that an error has occurred. No further calls should be made until the error has been corrected. Possible values are:
-1 One or more of the stated restrictions on the components $1 \leq$ matrix $\%$ class $\leq 4$, matrix $\% n \geq 0,0 \leq$ matrix $\% m$ $\leq$ matrix $\%$ m_max, $(0 \leq$ matrix $\%$ m $\leq$ matrix $\%$ m_max -1 in SCU_append $) 1 \leq$ col_del $\leq$ matrix $\% m$ and $1 \leq$ row_del $\leq$ matrix $\%$ m has been violated.
-2 The subroutine has been called with an initial value status $\leq 0$.
-3 The factors of $\mathbf{S}$ have not yet been formed in data. This indicates that either SCU_factorize has not yet been called, or that the last call to SCU_factorize, SCU_append or SCU_delete ended in a failure.
-4 One or more of the arrays matrix $\%$ BD_val, matrix $\%$ BD_row and matrix $\%$ BD_col_start has not been allocated.
-5 When the extended matrix is unsymmetric, one or more of the arrays matrix $\% C D \_v a l$, matrix $\% C D \_c o l$ and matrix $\%$ CD_row_start has not been allocated.
-6 One or more of the arrays matrix $\% \mathrm{BD}_{2}$ val, matrix $\% \mathrm{BD}_{\text {_row }}$ and matrix $\% \mathrm{BD}_{2}$ col_start is not large enough. Check that the dimension of matrix $\%$ BD_col_start is no smaller than matrix $\% \mathrm{~m}+1$ (matrix\%m+2 for SCU_append ), and that those of matrix $\%$ BD_val and matrix $\% \mathrm{BD}_{\text {_row }}$ are no smaller than matrix $\mathrm{BD}_{\mathrm{BD}}$ col_start (mat-rix\%m+1)-1, and re-enter. (matrix\%BD_col_start (matrix\%m+2)-1 for SCU_append and matrix $\%$ BD_col_start (matrix\%m+1) + |matrix\%col_del-matrix\%row_del|-1 for SCU_delete ).
-7 When the extended matrix is unsymmetric, one or more of the arrays matrix $\% C D \_v a l$, matrix $\% C D \_c o l$ and matrix\%CD_row_start is not large enough. Check that the dimension of matrix $\%$ CD_row_start is no smaller than matrix $\% \mathrm{~m}+1$ (matrix $\%$ +2 for SCU_append ), and that those of matrix $\% C D \_v a l$ and matrix $\% C D \_c o l$ are no smaller than matrix\%CD_row_start (matrix\%m+1)-1 (matrix\%CD_row_start (matrix\%m+2)-1 for SCU_append and matrix\%CD_row_start (matrix\%m+1) + |matrix\%col_del-matrix\%row_del|-1 for SCU_delete ).
-8 The value recorded in matrix_m does not correspond to the dimension of $\mathbf{D}$.
-9 The Schur complement matrix is singular; this has been detected during the $\mathbf{Q R}$ factorization of $\mathbf{S}$.
-10 The Schur complement matrix is expected to be positive definite, but this has been found not to be the case during the Cholesky factorization of $\mathbf{S}$.
-11 The Schur complement matrix is expected to be negative definite, but this has been found not to be the case during the Cholesky factorization of $-\mathbf{S}$.
-12 An internal array allocation or deallocation failed. See info\%alloc_status for further details.

### 2.6 Multiple updates

Once SCU_append or SCU_delete has been used to update the factorization of the Schur complement matrix, it is as if the enlarged or reduced matrix were that originally factorized by SCU_factorize. Consequently sequences of row and column additions and removals may be performed so long as sufficient storage is available.

## 3 GENERAL INFORMATION

Use of common: None.
Workspace: Provided automatically by the module.

## All use is subject to the conditions of a BSD-3-Clause License. <br> See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

Other routines called directly: the basic linear algebra subprograms (BLAS) SROT and SROTG (DROT and DROTG in GALAHAD_SCU_DOUBLE ) are called.

Other modules used directly: None.
Input/output: None.
Restrictions: matrix $\% \mathrm{n} \geq 0,0 \leq$ matrix $\% \mathrm{~m} \leq \operatorname{matrix} \% \mathrm{~m}$ _max and $1 \leq$ matrix $\%$ class $\leq 4$. Also $1 \leq$ row_del $\leq$ matrix $\%$ m and $1 \leq$ col_del $\leq$ matrix $\%$ m for SCU_delete.

Portability: ISO Fortran $95+$ TR 15581 or Fortran 2003. The package is thread-safe.

## 4 METHOD

The subroutine SCU_factorize forms the Schur complement $\mathbf{S}=\mathbf{D}-\mathbf{C A}^{-1} \mathbf{B}$ of $\mathbf{A}$ in the extended matrix by repeated reverse communication to obtain the columns of $\mathbf{A}^{-1} \mathbf{B}$. The Schur complement or its negative is then factorized into its QR or, if possible, Cholesky factors.

The subroutine SCU_solve solves the extended system using the following well-known scheme:
(i) Compute the solution to $\mathbf{A u}=\mathbf{b}_{1}$;
(ii) Compute $\mathbf{x}_{2}$ from $\mathbf{S x}_{2}=\mathbf{b}_{2}-\mathbf{C u}$;
(iii) Compute the solution to $\mathbf{A v}=\mathbf{B} \mathbf{x}_{2}$; and
(iv) Compute $\mathbf{x}_{1}=\mathbf{u}-\mathbf{v}$.

The subroutines SCU_append and SCU_delete compute the factorization of the Schur complement after a row and column have been appended to, and removed from, the extended matrix, respectively. The existing factorization is updated to obtain the new one; this is normally more efficient than forming the factorization from scratch.

## 5 EXAMPLE OF USE

As a simple example, suppose we wish to solve the system of equations

$$
\left(\begin{array}{lllllll}
1 & 0 & 0 & 0 & 0 & 1 & 0 \\
0 & 2 & 0 & 0 & 0 & 1 & 0 \\
0 & 0 & 3 & 0 & 0 & 1 & 0 \\
0 & 0 & 0 & 4 & 0 & 1 & 0 \\
0 & 0 & 0 & 0 & 5 & 1 & 1 \\
1 & 1 & 1 & 1 & 1 & 1 & 2 \\
1 & 0 & 1 & 0 & 1 & 3 & 4
\end{array}\right)\left(\begin{array}{l}
x_{1} \\
x_{2} \\
x_{3} \\
x_{4} \\
x_{5} \\
x_{6} \\
x_{7}
\end{array}\right)=\left(\begin{array}{c}
2 \\
3 \\
4 \\
5 \\
7 \\
8 \\
10
\end{array}\right) .
$$

Notice that the leading 5 by 5 coefficient matrix is diagonal and hence easily invertible. So we might choose $n=5$, $m=2$, and use SCU_factorize/SCU_solve to find the required solution. As the matrix is unsymmetric, we must set matrix\%class=1.

[^4]Now suppose that we have solved this system as described and are now confronted with the further system

$$
\left(\begin{array}{llllllll}
1 & 0 & 0 & 0 & 0 & 1 & 0 & 1 \\
0 & 2 & 0 & 0 & 0 & 1 & 0 & 0 \\
0 & 0 & 3 & 0 & 0 & 1 & 0 & 0 \\
0 & 0 & 0 & 4 & 0 & 1 & 0 & 0 \\
0 & 0 & 0 & 0 & 5 & 1 & 1 & 0 \\
1 & 1 & 1 & 1 & 1 & 1 & 2 & 1 \\
1 & 0 & 1 & 0 & 1 & 3 & 4 & 0 \\
1 & 0 & 0 & 0 & 0 & 0 & 1 & 1
\end{array}\right)\left(\begin{array}{l}
x_{1} \\
x_{2} \\
x_{3} \\
x_{4} \\
x_{5} \\
x_{6} \\
x_{7} \\
x_{8}
\end{array}\right)=\left(\begin{array}{c}
5 \\
5 \\
4 \\
5 \\
7 \\
12 \\
12 \\
4
\end{array}\right) .
$$

Rather than applying SCU_factorize /SCU_solve with $n=5, m=3$, we note that the new coefficient matrix differs from the old one merely in having an extra row and column. Thus, we can use SCU_append with $n=5, m=2$ to update the existing factorization and then call SCU_solve with $n=5, m=3$ to calculate the desired solution.

Finally, suppose that we have solved this second system and now wish to solve

$$
\left(\begin{array}{lllllll}
1 & 0 & 0 & 0 & 0 & 1 & 1 \\
0 & 2 & 0 & 0 & 0 & 1 & 0 \\
0 & 0 & 3 & 0 & 0 & 1 & 0 \\
0 & 0 & 0 & 4 & 0 & 1 & 0 \\
0 & 0 & 0 & 0 & 5 & 1 & 0 \\
1 & 0 & 1 & 0 & 1 & 3 & 0 \\
1 & 0 & 0 & 0 & 0 & 0 & 1
\end{array}\right)\left(\begin{array}{l}
x_{1} \\
x_{2} \\
x_{3} \\
x_{4} \\
x_{5} \\
x_{6} \\
x_{7}
\end{array}\right)=\left(\begin{array}{l}
3 \\
5 \\
4 \\
5 \\
6 \\
6 \\
2
\end{array}\right) .
$$

Again, rather than applying SCU_factorize /SCU_solve with $n=5, m=2$, we note that this third coefficient matrix is the second one with its $n+1$-st row and $n+2$-nd column removed. Thus, we can use SCU_delete with $n=5, m=3$ to update the existing factorization and then call SCU_solve with $n=5, m=2$ to calculate the desired solution.

To carry out these calculations, we might use the following piece of code. Notice how the extra row and column for the second problem are simply introduced at the end of the existing data structures.

```
PROGRAM GALAHAD_SCU_EXAMPLE
USE GALAHAD_SCU_DOUBLE ! double precision version
IMPLICIT NONE
INTEGER, PARAMETER :: wp = KIND( 1.OD+0 ) ! set precision
INTEGER :: i, row_del, col_del, status
INTEGER, PARAMETER :: n = 5, m = 2, mmax = m + 1
TYPE ( SCU_matrix_type ) :: mat
TYPE ( SCU_data_type ) :: data
TYPE ( SCU_info_type ) :: info
REAL ( KIND = wp ) :: X1 ( n + m ), RHS1( n + m )
REAL ( KIND = wp ) : : X2( n + m + 1 ), RHS2( n +m + 1)
REAL ( KIND = wp ) :: X3( n + m ), RHS3( n + m ), VECTOR( n )
mat%m_max = mmax ; mat%class = 1
mat%n = n ; mat%m = m
ALLOCATE ( mat%BD_val( 15 ), mat%BD_row( 15 ), mat%BD_col_start( mmax+1 ),&
    mat%CD_val( 13 ), mat%CD_col( 13 ), mat%CD_row_start( mmax+1 ) )
mat%BD_col_start( : 3 ) = (/1, 7, 10/)
mat%CD_row_start( : 3 ) = (/1, 6, 10/)
mat%BD_row ( : 9 ) = (/1, 2, 3, 4, 5, 6, 5, 6, 7/)
mat%BD_val( : 9 ) = (/1.0_wp, 1.0_wp, 1.0_wp, 1.0_wp, &
    1.0_wp, 1.0_wp, 1.0_wp, 2.0_wp, 4.0_wp/)
mat%CD_col( : 9 ) = (/1, 2, 3, 4, 5, 1, 3, 5, 6/)
```


## All use is subject to the conditions of a BSD-3-Clause License.

See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

```
mat%CD_val( : 9 ) = (/1.0_wp, 1.0_wp, 1.0_wp, 1.0_wp, &
    1.0_wp, 1.0_wp, 1.0_wp, 1.0_wp, 3.0_wp/)
RHS1 = (/2.0_wp, 3.0_wp, 4.0_wp, 5.0_wp, 7.0_wp, 8.0_wp, 10.0_wp/)
RHS2 = (/5.0_wp, 5.0_wp, 4.0_wp, 5.0_wp, 7.0_wp, 12.0_wp, 12.0_wp, 4.0_wp/)
RHS3 = (/3.0_wp, 5.0_wp, 4.0_wp, 5.0_wp, 6.0_wp, 6.0_wp, 2.0_wp/)
First system
status = 1
DO
    CALL SCU_factorize( mat, data, VECTOR, status, info )
    IF ( status <= 0 ) EXIT
    DO i = 1, n
        VECTOR( i ) = VECTOR( i ) / DBLE ( FLOAT ( i ) )
    END DO
END DO
WRITE( 6, "( /, ' On exit from SCU_factorize, status = ', I3 )" ) status
IF ( status < 0 ) STOP
status = 1
DO
    CALL SCU_solve( mat, data, RHS1, X1, VECTOR, status )
    IF ( status <= 0 ) EXIT
    DO i = 1, n ! multiply by the inverse of A
        VECTOR( i ) = VECTOR( i ) / DBLE( FLOAT ( i ) )
    END DO
END DO
WRITE( 6, "( ' On exit from SCU_solve, status = ', I3 )" ) status
IF ( status < 0 ) STOP
WRITE( 6, "( /, ' Solution (first system)', /, ( 8ES9.2 ) )" ) X1( : )
! Second system
mat%BD_row( 10 : 12 ) = (/ 1, 6, 8 /)
mat%BD_val( 10 : 12 ) = (/ 1.0_wp, 1.0_wp, 1.0_wp /)
mat%BD_col_start( 4 ) = 13
mat%CD_col( 10 ) = 1
mat%CD_val( 10 ) = 1.0_wp
mat%CD_row_start( 4 ) = 11
status = 1
DO
    CALL SCU_append( mat, data, VECTOR, status, info )
    IF ( status <= 0 ) EXIT
    DO i = 1, n
        VECTOR( i ) = VECTOR( i ) / DBLE( FLOAT( i ) )
    END DO
END DO
WRITE( 6, "( /, ' On exit from SCU_append, status = ', I3 )" ) status
IF ( status < 0 ) STOP
status = 1
DO
    CALL SCU_solve( mat, data, RHS2, X2, VECTOR, status )
    IF ( status <= 0 ) EXIT
    DO i = 1, n
        VECTOR( i ) = VECTOR( i ) / DBLE( FLOAT ( i ) )
    END DO
```


## All use is subject to the conditions of a BSD-3-Clause License. <br> See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

```
END DO
WRITE( 6, "( ' On exit from SCU_solve, status = ', I3 )" ) status
IF ( status < 0 ) STOP
WRITE( 6, "( /, ' Solution (second system)', /, ( 8ES9.2 ) )" ) X2
! Third system
row_del = 1
col_del = 2
status = 1
CALL SCU_delete( mat, data, VECTOR, status, info, col_del, &
    row_del = row_del )
WRITE( 6, "( /, ' On exit from SCU_delete, status = ', I3 )" ) status
IF ( status < 0 ) STOP
status = 1
DO
    CALL SCU_solve( mat, data, RHS3, X3, VECTOR, status )
    IF ( status <= 0 ) EXIT
                DO i = 1, n
                VECTOR( i ) = VECTOR( i ) / DBLE( FLOAT( i ) )
    END DO
END DO
WRITE( 6, "( ' On exit from SCU_solve, status = ', I3 )" ) status
IF ( status < 0 ) STOP
WRITE( 6, "( /, ' Solution (third system)', /, ( 8ES9.2 ) )" ) X3
CALL SCU_terminate( data, status, info )
WRITE( 6, "( /, ' On exit from SCU_terminate, status = ', I3 )" ) status
END PROGRAM GALAHAD_SCU_EXAMPLE
```

This produces the following output:

```
On exit from SCU_factorize, status = 0
On exit from SCU_solve, status = 0
Solution (first system)
1.00E+00 1.00E+00 1.00E+00 1.00E+00 1.00E+00 1.00E+00 1.00E+00
On exit from SCU_append, status = 0
On exit from SCU_solve, status = 0
Solution (second system)
3.00E+00 2.00E+00 1.00E+00 1.00E +00 1.00E+00 1.00E+00 1.00E+00 1.00E+00
On exit from SCU_delete, status = 0
On exit from SCU_solve, status = 0
Solution (third system)
1.00E+00 2.00E+00 1.00E+00 1.00E+00 1.00E+00 1.00E+00 1.00E+00
On exit from SCU_terminate, status = 0
```


[^0]:    All use is subject to the conditions of a BSD-3-Clause License.
    See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

[^1]:    All use is subject to the conditions of a BSD-3-Clause License.
    See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

[^2]:    All use is subject to the conditions of a BSD-3-Clause License.
    See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

[^3]:    All use is subject to the conditions of a BSD-3-Clause License.
    See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

[^4]:    All use is subject to the conditions of a BSD-3-Clause License.
    See http://galahad.rl.ac.uk/galahad-www/cou.html for full details.

